Master Mind program for the HP-41C family

A.J.M. Jacobs

March 5, 2013

Contents

1 The Master Mind program HP-41C code

key strokes	step	display code(s)	remark
[ON]			Put the calculator [ON]
XEQ [ALPHA]SIZE[ALPHA] 25			Set the minumum register size
[PRGM]			Enter program mode
□ GTO . .		00 REG nnn	Set program counter @ end of code
□ LBL [ALPHA]GMM[ALPHA]	01	LBL"GMM	Start position code Generate Master Mind secret code
RCL 19	02	RCL 19	Read random-seed value
.1	03	.1	
+	04	+	Add .1 to the seed
XEQ [ALPHA]ABS[ALPHA]	05	ABS	Make sure this value is positive
STO 19	06	STO 19	Save a new random-seed value
0	07	0	
STO 11	08	STO 11	Reset the secret code
STO 12	09	STO 12	
STO 13	10	STO 13	
□ LBL 01	11	LBL 01	Loop label generating code by code
XEQ 09	12	XEQ 09	Call pseudo random function
STO 14	13	STO 14	Store the (first) random code number
RCL 11	14	RCL 11	Check if code is unique
□ X=Y?	15	X=Y?	
□ GTO 01	16	GTO 01	No? Generate a new code
XEQ [ALPHA]X □ H □ 0 ?[ALPHA]	17	X≠0?	Is there all ready a code?

key strokes	step	display code(s)	remark
□ GTO 02	18	GTO 02	No? Move on
RCL 14	19	RCL 14	
STO 11	20	STO 11	Store the secret code
□ GTO 01	21	GTO 01	Loop for the second
□ LBL 02	22	LBL 02	Next code
RCL 14	23	RCL 14	
RCL 12	24	RCL 12	
□ X=Y?	25	X=Y?	Is the code unique?
□ GTO 01	26	GTO 01	No? Generate a new code
XEQ [ALPHA]X □ H □ 0 ?[ALPHA]	27	X≠0?	Is there all ready a code?
□ GTO 03	28	GTO 03	No? Move on
RCL 14	29	RCL 14	
STO 12	30	STO 12	Store the secret code
□ GTO 01	31	GTO 01	Loop for a new code
□ LBL 03	32	LBL 03	
RCL 14	33	RCL 14	
RCL 13	34	RCL 13	
□ X=Y?	35	X=Y?	Is the code unique?
□ GTO 01	36	GTO 01	No? Generate a new code
XEQ [ALPHA]X □ H □ 0 ?[ALPHA]	37	X≠0?	Is there all ready a code?
□ GTO 04	38	GTO 04	No? Move on
RCL 14	39	RCL 14	
STO 13	40	STO 13	Store the secret code

key strokes	step	display code(s)	remark
□ GTO 01	41	GTO 01	Generate the last code
□ LBL 04	42	LBL 04	
0	43	0	Clear the display
□ RTN	44	RTN	Return
□ LBL 09	45	LBL 09	The pseudo random number generator
RCL 19	46	RCL 19	Read previous pseudo random number
LN	47	LN	LN * 100
EEX 2	48	1 E2	
×	49	*	
1	50	1	
XEQ [ALPHA]MOD[ALPHA]	51	MOD	Modulo (1) Number from 0 till 1
STO 19	52	STO 19	Store the new pseudo random number
6	53	6	Create a number from 1 till 6
×	54	*	
1	55	1	
+	56	+	
XEQ [ALPHA]INT[ALPHA]	57	INT	Integer 1..6
□ RTN	58	RTN	Return

key strokes	step	display code(s)	remark
□ LBL [ALPHA]MM[ALPHA]	59	LBL"MM	Start position code memory test
□ FIX 1	60	FIX 1	Set the display format to 0.0
STO 20	61	STO 20	Store the users guess in register 20
10	62	10	
×	63	*	
STO 21	64	STO 21	Store the users guess in register 21
XEQ 10	65	XEQ 10	Get lowest digit
STO 24	66	STO 24	Store the last number of the code in register 24
XEQ 10	67	XEQ 10	Get lowest digit
STO 23	68	STO 23	Store the third number of the code in register 23
XEQ 10	69	XEQ 10	Get lowest digit
STO 22	70	STO 22	Store the second number of the code in register 22
XEQ 10	71	XEQ 10	Get last digit, saved in register 21
0	72	0	
STO 00	73	STO 00	Set the score to 0
XEQ 07	74	XEQ 07	Call score count routine
4	75	4	
□ X=Y?	76	X=Y?	Is the score a 4.0?
□ GTO 05	77	GTO 05	The user cracked the code
RCL 00	78	RCL 00	Read the score, hits and near hits
□ RTN	79	RTN	Return

key strokes	step	display code(s)	remark
□ LBL 10	80	LBL 10	Help function, integer division and lowest digit
RCL 21	81	RCL 21	Read value
10	82	10	
÷	83	/	Divide by 10
XEQ [ALPHA]INT[ALPHA]	84	INT	Round to integer
STO 21	85	STO 21	Store value
10	86	10	
÷	87	/	
XEQ [ALPHA]FRC[ALPHA]	88	FRC	Calculate the lowest digit
10	89	10	
×	90	*	
□ RTN	91	RTN	Return
□ LBL 05	92	LBL 05	The user cracked the code
□ FIX 0	93	FIX 0	Set display format to 0.
RCL 20	94	RCL 20	Read the secret code
□ BEEP	95	BEEP	Give a beep
□ RTN	96	RTN	Return
□ LBL 07	97	LBL 07	Routine to count the point
RCL 21	98	RCL 21	First check right digit on wrong position
RCL 12	99	RCL 12	Simple brute force
□ X=Y?	100	X=Y?	Compare guess with secret code
XEQ 08	101	XEQ 08	Add 1 to score
RCL 21	102	RCL 21	

key strokes	step	display code(s)	remark
RCL 13	103	RCL 13	
□ X=Y?	104	X=Y?	Compare guess with secret code
XEQ 08	105	XEQ 08	Add 1 to score
RCL 21	106	RCL 21	
RCL 14	107	RCL 14	
□ X=Y?	108	X=Y?	Compare guess with secret code
XEQ 08	109	XEQ 08	Add 1 to score
RCL 22	110	RCL 22	
RCL 11	111	RCL 11	
□ X=Y?	112	X=Y?	Compare guess with secret code
XEQ 08	113	XEQ 08	Add 1 to score
RCL 22	114	RCL 22	
RCL 13	115	RCL 13	
□ X=Y?	116	X=Y?	Compare guess with secret code
XEQ 08	117	XEQ 08	Add 1 to score
RCL 22	118	RCL 22	
RCL 14	119	RCL 14	
□ X=Y?	120	X=Y?	Compare guess with secret code
XEQ 08	121	XEQ 08	Add 1 to score
RCL 23	122	RCL 23	
RCL 11	123	RCL 11	
□ X=Y?	124	X=Y?	Compare guess with secret code
XEQ 08	125	XEQ 08	Add 1 to score
RCL 23	126	RCL 23	
RCL 12	127	RCL 12	

key strokes	step	display code(s)	remark
□ X=Y?	128	X=Y?	Compare guess with secret code
XEQ 08	129	XEQ 08	Add 1 to score
RCL 23	130	RCL 23	
RCL 14	131	RCL 14	
□ X=Y?	132	X=Y?	Compare guess with secret code
XEQ 08	133	XEQ 08	Add 1 to score
RCL 24	134	RCL 24	
RCL 11	135	RCL 11	
□ X=Y?	136	X=Y?	Compare guess with secret code
XEQ 08	137	XEQ 08	Add 1 to score
RCL 24	138	RCL 24	
RCL 12	139	RCL 12	
□ X=Y?	140	X=Y?	Compare guess with secret code
XEQ 08	141	XEQ 08	Add 1 to score
RCL 24	142	RCL 24	
RCL 13	143	RCL 13	
□ X=Y?	144	X=Y?	Compare guess with secret code
XEQ 08	145	XEQ 08	Add 1 to score
10	146	10	
STO ÷ 00	147	ST/ 00	Save score, number after decimal point
RCL 11	148	RCL 11	Next check right digit on right position
RCL 21	149	RCL 21	Simple brute force
□ X=Y?	150	X=Y?	Compare guess with secret code
XEQ 08	151	XEQ 08	Add 1 to score
RCL 12	152	RCL 12	

key strokes	step	display code(s)	remark
RCL 22	153	RCL 22	
□ X=Y?	154	X=Y?	Compare guess with secret code
XEQ 08	155	XEQ 08	Add 1 to score
RCL 13	156	RCL 13	
RCL 23	157	RCL 23	
□ X=Y?	158	X=Y?	Compare guess with secret code
XEQ 08	159	XEQ 08	Add 1 to score
RCL 14	160	RCL 14	
RCL 24	161	RCL 24	
□ X=Y?	162	X=Y?	Compare guess with secret code
XEQ 08	163	XEQ 08	Add 1 to score
RCL 00	164	RCL 00	Get the total score
□ RTN	165	RTN	Return
□ LBL 08	166	LBL 08	Increment counter routine
1	167	1	Go add 1
STO + 00	168	ST+ 00	Store in register 00
□ RTN	169	RTN	Return
□ GTO . .		00 REG nnn	End RPN coding
[PRGM]			Leave program mode
□ ASN [ALPHA]GMM[ALPHA] Σ+			Assign "GMM" to Σ+
□ ASN [ALPHA]MM[ALPHA] LN			Assign "MM" to LN
[USER]			Set USER mode

2 How to use the Master Mind program

With routine "GMM" the calculator will generate a secret code in the form ABCD. A, B, C and D are (pseudo) random numbers from one to six. A, B, C and D are not equal to each other.

The RPN coded Master Mind program with the Labels "GMM" and "MM" have been assigned to keys Σ+ and LN. The HP-41C calculator has been set in USER-mode.

Example game:

Keystrokes: Σ+

Running... (a secret code is created)

Enter your first guess:

`1234`

`LN`

Output: `2.1`

Two numbers are correct and at the right place, one number is correct but at the wrong place.

Enter your second guess:

`1253`

`LN`

Output: `1.3`

One number is correct and at the right place, three numbers are correct but at the wrong place. So the four numbers are all numbers in de code.

Enter your third guess:

`5231`

`LN`

Output: `1.3`

One number is correct and at the right place, three numbers are correct but at the wrong place.

Enter your fourth guess:

`3215`

`LN`

Output: `0.4`

All the numbers are correct but at the wrong place.

Enter your fifth guess:

`1532`

`LN`

Output: (beep) `1,532.`

You cracked the secret code.